Lab nr 7

Dodatek

Program zajęć:

1. **Propozycja rozwiązania zadania z Lab nr 6**
   a. Dla URA z regulatorem rzeczywistym typu P $\left(G_r(s) = \frac{k_p}{T_1 s+1}\right)$, obiektem całkującym rzeczywistym $\left(G_{ob}(s) = \frac{k}{s(T_2 s+1)}\right)$ oraz idealnym czujnikiem pomiarowym, wyznaczyć zakresy parametrów k_p, k, T_1 i T_2, przy których URA będzie stabilny oraz będzie na granicy stabilności.

   Proponowany sposób rozwiązania:

      i. wyznaczyć transmitancję układu otwartego,
      ii. wyznaczyć równanie charakterystyczne,
      iii. n=3, zatem zbadać wyznacznik drugiego stopnia (wg. kryterium Hurwitza) – wzajemna relacja stałych czasowych i współczynników wzmocnień wskaże zakresy parametrów.

   b. Sprawdzić rozwiązania z wykorzystaniem Scilab:
      i. sprawdzić położenie biegunów,
      ii. sprawdzić charakterystyki Nyquista dla układów otwartych,
      iii. sprawdzić reakcję URA na pobudzenie impulsowe dla wyznaczonych zakresów parametrów (Scilab lub/i Xcos),
      iv. sprawdzić odpowiedzi skokowe dla URA.

ROZWIĄZANIE:



Transmitancja układu otwartego: $G_o(s) = G_r(s) \cdot G_{ob}(s) = \frac{k_p}{T_1 s+1} \cdot \frac{k}{s(T_2 s+1)}$

Transmitancja układu zamkniętego:

$$G_z(s) = \frac{G_o(s)}{1+G_o(s)} = \frac{G_r(s) \cdot G_{ob}(s)}{1+G_r(s) \cdot G_{ob}(s)} = \cdots = \frac{k k_p}{s(T_1 s+1)(T_2 s+1)+k k_p}$$

Równanie charakterystyczne: $s(T_1 s + 1)(T_2 s + 1) + kk_p = 0$

Po przekształceniach: $T_1 T_2 s^3 + (T_1 + T_2)s^2 + s + kk_p = 0$

n=3, zatem zbadamy wyznacznik drugiego stopnia:

$\Delta_2 = \begin{vmatrix} a_2 & a_0 \\ a_3 & a_1 \end{vmatrix} = \begin{vmatrix} T_1 + T_2 & kk_p \\ T_1 T_2 & 1 \end{vmatrix} = T_1 + T_2 - kk_p T_1 T_2$

Aby URA był **stabilny**, parametry (nastawy) regulatora k_p i T_1 oraz parametry obiektu k i T_2 muszą spełniać warunek:

$\Delta_2 > 0 \rightarrow T_1 + T_2 - kk_p T_1 T_2 > 0$

$$\frac{1}{T_1} + \frac{1}{T_2} > kk_p$$

Aby URA był **na granicy stabilności**:

$$\frac{1}{T_1} + \frac{1}{T_2} = kk_p$$

Sprawdzenie dla przypadku stabilnego (dodatkowy indeks dolny $_1$):

a) stabilny: $T_{11}=2$; $T_{21}=3$; $k_1=0.5$; $k_{p1}=0.5$

Sprawdzenie dla przypadku na granicy stabilności (dodatkowy indeks dolny $_2$):

b) na granicy stabilności: $T_{12}=2$; $T_{22}=3$; $k_2=5/6$; $k_{p2}=1$

```
// Lab7
s=%s;
// podpunkt a
k1=0.5;
T21=3;
Gob1=syslin('c',k1/((T21*s*s)+s)); // tworzy transmitancję obiektu1
disp('transmitancja obiektu 1:');
disp(Gob1);
t=0:0.1:100;
Gcz=syslin('c',1,1); // tworzy transmitancję czujnika w torze pętli sprzężenia zwrotnego
disp('transmitancja czunika:');
disp(Gcz);
kp1=0.5; // współczynnik wzmocnienia regulatora1 - rzeczywistego typu P
T11=2; // stała czasowa regulatora rzeczywistego typu P
Gr1=syslin('c',kp1/((T11*s)+1)); //tworzy transmitancję regulatora1
disp('transmitancja regulatora 1:');
disp(Gr1);
Gz1=(Gr1*Gob1)/.Gcz; // tworzy transmitancję układu zamkniętego (URA1)
disp('transmitancja URA1:');
disp(Gz1);
y1=csim('step',t,Gz1); // generuje odpowiedz skokową URA1
subplot(2,2,1);
plot(t,y1);
subplot(2,2,2);
plzr(Gz1); // wykreśla zera i bieguny URA1
// podpunkt b
k2=5/6;
T22=3;
Gob2=syslin('c',k2/((T22*s*s)+s)); // tworzy transmitancję obiektu2
disp('transmitancja obiektu 2:');
disp(Gob2);
kp2=1;
T12=2;
Gr2=syslin('c',kp2/((T12*s)+1)); //tworzy transmitancję regulatora2
disp('transmitancja regulatora 2:');
disp(Gr2);
Gz2=(Gr2*Gob2)/.Gcz;
disp('transmitancja URA2:');
disp(Gz2);
y2=csim('step',t,Gz2);
subplot(2,2,3);
plot(t,y2);
subplot(2,2,4);
plzr(Gz2);
m1=[6 5 1 0.25]; //deklaracja współczynników wielomianu mianownika URA1 (wielomianu
charakterystycznego) - wartości odczytane z wcześniejszych wyników działania skryptu
x1=roots(m1); //wyznaczenie pierwiastków mianownika (biegunów) URA1
disp('bieguny URA1:');
disp(x1); //wyświetlenie wartości biegunów URA1
m2=[6 5 1 0.8333333333333333]; // wartości odczytane z wcześniejszych wyników działania
skryptu
x2=roots(m2);
disp('bieguny URA2:');
disp(x2);
```



Rys. 1. Odpowiedzi skokowe i położenie biegunów układów URA1 i URA2.

"bieguny URA1 (pierwiastki równania charakterystycznego):"
-0.6781705 + 0.i
-0.0775814 + 0.2354165i
-0.0775814 - 0.2354165i

" bieguny URA2 (pierwiastki równania charakterystycznego):"
-0.8333333 + 0.i
-6.245D-17 + 0.4082483i (tutaj część rzeczywista jest = $-6.245 \cdot 10^{-17} = \sim 0$)
-6.245D-17 - 0.4082483i (tutaj część rzeczywista jest = $-6.245 \cdot 10^{-17} = \sim 0$)

Przykłady URA

źródło: Dorf R. C., Bishop R. H., *MODERN CONTROL SYSTEMS Solution Manual*, Prentice Hall, New Jersey 2011 [on-line]

https://wp.kntu.ac.ir/dfard/ebook/lc/[Richard_C._Dorf,_Robert_H._Bishop]_Instructor%27s_S(b-ok.org).pdf

http://web.iaa.ncku.edu.tw/~chiehli/course/control/Homeworks%202-10%20Dorf%2012ed.pdf